
1Introduction

Powerful yet easy-to-use, MKS awk is the perfect programming language for 
anyone who needs to work with information stored in text files. The general 
purpose of awk is to read the contents of one or more files, obtain selected 
pieces of information from the file(s), and present the information in a 
specified format. By default, the information presented will be displayed on 
your screen; however, it can be written into a file with the usual >filename 
redirection construct. 

awk is a good first programming language. It allows most of the logical 
constructs of modern computing languages: if-else statements, while 
and for loops, function calls, and so on. It is easy to learn, and lets you get 
results with little effort. At the same time, it introduces all the primary 
concepts of programming and prepares you for more complicated languages. 

This document introduces the important principles and concepts of MKS awk, 
and demonstrates how you can use them for productive programming. For 
beginners, this chapter should serve as a good introduction; experienced 
programmers might prefer the detailed technical information of the online 
awk reference page. 

2Basic Concepts

Every programming language has its own way of looking at the world. To 
write programs in the language, you must learn to see things from the 
language's point of view. 

This section examines the fundamentals of awk:

 the kind of information, or data, awk works with;

 the shape of an awk program;

 how to run an awk program.

Data and Data Files
awk programs work with data. Programs can obtain data typed in from the 
terminal or from the output of other commands (for example, through pipes); 
but usually data come from data files. awk's data files are always text files. 
This means that the files contain readable text: for example, words, numbers, 
punctuation characters, and so on. 

As an example, consider a data file containing information on the hobbies of 
a group of people. Each line in this file gives a person's name, one of that 
person's hobbies, how many hours a week that person spends on the hobby, 

and how much money the hobby costs per year. Each of a person's hobbies 
appears on a separate line. We’ve included just such a file called hobbies 
installed in the ROOTDIR/samples/guide directory:

Jim     reading          15     100.00
Jim     bridge            4      10.00
Jim     role-playing      5      70.00
Linda   bridge           12      30.00
Linda   cartooning        5      75.00
Katie   jogging          14     120.00
Katie   reading          10      60.00
John    role-playing      8     100.00
John    jogging           8      30.00
Andrew  wind-surfing     20    1000.00
Lori    jogging           5      30.00
Lori    weight-lifting   12     200.00
Lori    bridge            2       0.00

Records An awk data file is a collection of records. A record contains a number of 
pieces of information about a single item; these pieces are called fields. In 
our hobbies file, each line is a separate record, giving a complete set of 
information about one of a person's hobbies.

Records are separated by a record separator character, which is usually the 
newline character for awk. A newline character shows where one line of text 
ends and another begins; by using the newline as a record separator, each 
line of the file becomes a separate record. For convenience we will use a 
newline character as a record separator in all of the following examples.

Fields A record consists of a number of fields, each of which contain a single piece 
of information. For example, the hobby record

Jim     reading          15     100.00

contains four fields:
Jim
reading
15
100.00

Fields should be provided in the same order in each record. That way awk 
and other programs can easily access a particular piece of information in any 
record. 

You can easily change this 
default, as discussed in “The awk 
Command Line” on page 10 and 
“Built-In String Variables” on 
page 50.

The fields of a record are separated by one or more field separator 
characters. The hobbies file uses strings of blank characters (spaces) to 
separate fields. By default, awk uses blanks and/or horizontal tab characters 
to separate fields. 

The Shape of a Program
An awk program looks like this:

pattern {actions}
pattern {actions}
pattern {actions}
...

Each line is a separate instruction or rule. awk looks through the data files 
record-by-record and executes the rules, in the given order, on each record. 

Note that awk programs can only be run from the command line when 
running the KornShell. If you want to run an awk program while operating 
under command.com or cmd.exe, use the -f option as in

awk -f file

or run the program from within a file.

Simple Patterns A rule of the form
pattern {actions}

indicates that awk is to perform the given set of actions on every record that 
meets a certain set of conditions. The conditions are given by the pattern part 
of the rule. 

The pattern of a rule often looks for records that have a particular value in 
some field. The notation $1 stands for the first field of a record, $2 stands 
for the second field, and so on. As a special case, $0 stands for the entire 
record. For example, here's a simple awk rule:

$2 == "jogging" { print }

The notation == stands for is equal to; therefore, the rule means: If the 
second field in a record is jogging, display the entire record.

This rule is a complete awk program. When you run this program on the 
hobbies file, awk looks through the file record-by-record (line-by-line). 
Whenever a line has jogging as its second field, awk displays the complete 
record. The output from this program therefore is

Katie   jogging          14     120.00
John    jogging           8      30.00
Lori    jogging           5      30.00

Let's take another example. Ask yourself what the following awk program 
does.

$1 == "John" { print }

As you probably guessed, it displays every record that has John as its first 
field. The output from the program is

John    role-playing      8     100.00
John    jogging           8      30.00

You can perform the same sort of search on any text database. The only 
difference is that databases tend to contain a great deal more data than this 
example. 

These examples both use the print action; however, this action never needs 
to be written explicitly. You could write the programs as

$2 == "jogging"

 and
$1 == "John"

and they would have exactly the same effect.
If an awk rule does not contain an 
action, print is assumed.

The == notation is an example of a comparison operation. awk recognizes 
several other types of comparison:

!= not equal
> greater than
< less than
>= greater than or equal to
<= less than or equal to

As examples, consider each of the following rules as complete programs, 
and decide for yourself what the programs will do with the hobbies file.

(i)    $1 != "Linda" { print }
(ii)   $3 > 10
(iii)  $4 < 100.00
(iv)   $4 <= 100.00



The Shape of a Program

Using MKS AWK 7

These rules have the following effects.

Numbers and 
Strings

In the previous examples, we put quote (") characters around Linda in (i), but 
did not use quotes in any of the other rules. awk distinguishes between string 
values (which are enclosed in quotes) and numeric values (which are not).

A string value is just a sequence of characters like "abc". Any sort of 
characters are allowed (even digits, as in "abc123"). Strings can contain any 
number of characters. A string with zero characters is called the null string 
and is written "".

awk lets you use exponential or 
scientific notation. Exponents are 
given as e or E, followed by an 
optionally signed exponent. 
Thus, 1E3, 1.0e3, 10E2, and 
1000 are all equivalent

A numeric value consists of digits, but it can also have a sign and a decimal 
point. For example,

10    0.34    -78    +2.56    -.92

are all valid numbers in awk. awk does not let you put commas inside 
numbers. For example, you must write 1000 instead of 1,000.

When awk compares numbers (with operators like > or <), it makes 
comparisons in accordance with the usual rules of arithmetic. When awk 
compares strings, it makes comparisons in accordance with the ASCII 
collating order. This is a little like alphabetical order; for example, the 
program

$1 >= "Katie"

displays the Katie, Linda, and Lori lines, which is what you would expect 
from alphabetical order; however, ASCII collating order differs from 
alphabetical order in a number of respects. For example, lowercase letters 
are greater than uppercase ones, so that a is greater than Z.

The Print Action So far, print has been the only action discussed. As you have seen, print 
can display an entire record. It can also display selected fields of the record, 
as in 

$2 == "bridge" { print $1 }

(i) Displays all records with a first field that is not Linda.

(ii) Displays all records with a third field that is greater than 10. 
Remember that when there is no explicit action, awk 
assumes print.

(iii) Displays all records with a fourth field that is less than 
100.00.

(iv) Displays all records with a fourth field that is less than or 
equal to 100.00.

This displays the first field of every record with a second field that is bridge. 
The output is

Jim
Linda
Lori

print can display more than one field. If you give print a list of fields 
separated by commas, as in

$1 == "Jim" { print $2,$3,$4 }

print displays the given fields separated by single blanks, as in:
reading 15 100.00
bridge 4 10.00
role-playing 5 70.00

The print action can display strings and numbers along with fields. For 
example,

$1 == "John" { print "$",$4 }

displays
$ 100.00
$ 30.00

In this rule, the print action displays a string containing a $, followed by a 
blank, followed by the value of the fourth field in each selected record. 

As an exercise, predict the output of the following:
(i)   $1 == "Lori" {print $1,"spends $", $4,"on",$2}
(ii)  $2 == "jogging" {print $1,"jogs",$3,"hours a week"}
(iii) $4 > 100.00 {print $1, "has an expensive hobby"}

Miscellaneous 
Points

 You can put any number of extra blanks and/or horizontal tabs into awk 
patterns and actions. For example, you can say:

{ print $1 , $2 , $3 }

 You can leave out the pattern part of a rule. In this case, awk applies the 
action part of the rule to every record in the file. For example, 

{ print }

is a complete awk program that displays every record in the data file.

 You can leave out the action part of a rule. In this case, the default action 
is print. For example,

$1 == "Andrew"

is a complete awk program that displays every record with a first field 
that is Andrew. This is equivalent to:

$1=="Andrew" { print }

 When an awk program contains several rules, awk applies every 
appropriate rule to the first record, then every appropriate rule to the 
second record, and so on. Rules are applied in order. For example, 
consider the following awk program.

$1 == "Linda"
$2 == "bridge" { print $1 }

The output of this program is:
Jim
Linda   bridge           12      30.00
Linda
Linda   cartooning        5      75.00
Lori

awk looks through the file record-by-record. The first record to satisfy 
one of the patterns is

Jim     bridge            4      10.00

so awk displays the first field of the record (as dictated by the second 
rule).

The next record of interest is
Linda   bridge           12      30.00

This satisfies the pattern of the first rule, so the whole record is 
displayed. It also satisfies the pattern of the second rule, so the first field 
is displayed. awk continues through the file, record-by-record, 
performing the appropriate actions when a record satisfies the pattern.

Running awk Programs
There are two ways to run awk programs: from a command line, and from a 
program file.

The awk 
Command Line

The simplest awk command line is
awk 'program' datafile

The awk program is enclosed in single quote/apostrophe (') characters. The 
datafile argument gives the name of the data file. For example,

awk '$1 == "Linda"' hobbies

runs the program
$1 == "Linda"

on the data file hobbies.

If you are using the MKS KornShell (not command.com or cmd.exe), you 
can type in a multi-line program within single quotes, as in: 

awk '
   $1 == "Linda"
   $2 == "bridge" { print $1 }
   ' hobbies

For more information on entering 
multi-line arguments, see the 
Using the MKS KornShell 
document available from the 
Start menu.

As mentioned previously, awk assumes that blanks and/or horizontal tabs 
separate fields in a record. If the data file uses different field separator 
characters, you must indicate this on the command line. You can do this with 
an option of the form

-Fstring

where the string lists the characters used to separate fields. For example,
awk -F":" '{ print $3 }' file.dat

indicates that the given data file uses colon (:) characters to separate record 
fields. The -F option must come before the quoted program rules. 

awk also allows you to define the value of variables on the command line by 
using the -v option. If you want to set the value of a variable foo before the 
program starts, then use the following command line:

awk -v foo=13 -f program datafile

Program Files Short programs like the ones in this chapter can be entered on a single 
command line. Later chapters discuss longer programs which you cannot 
type on a single line. Such programs are most easily run from a program file.

A program file is a (text) file that contains an awk program. You can create 
program files with any text editor (such as vi). For example, you might 
create a file named lbprog.awk that contains the lines:

$1 == "Linda"
$2 == "bridge" { print $1 }

To run a program on a particular data file, use the command
awk -f progfile datafile

where progfile is the name of the file that contains the awk program and 
datafile is the name of the data file. For example,

awk -f lbprog.awk hobbies

runs the program in lbprog.awk on the data in hobbies.

If the data file does not use the default separator characters, you must specify 
a -F option after the progfile name, as in 

awk -f prog.awk -F":" file.dat

To gain some experience using awk you should run the examples of this 
chapter on the hobbies file. Run some from the command line and some 
from program files. 

Sources of Data If you do not specify a data file on the command line, awk reads data from 
the terminal. For example, if you issue the command

awk '{ print $1 }'

awk displays the first word of every line you type. When you type in data 
from the terminal, you can mark the end of the data by typing CTRL-Z. After 
you type CTRL-Z, you must press ENTER.

A command line may also specify several data files, as in
awk -f progfile data1 data2 data3 ...

When awk has finished reading through the first data file data1, it goes 
on to data2, and so on.

3Simple Arithmetic

With awk you can easily perform calculations with numbers contained in 
data files. This chapter discusses how awk does arithmetic and gives some 
examples of programs using these features.

Note that awk, with minor differences, performs arithmetic operations in 
much the same way as the C programming language; therefore, knowledge 
of awk is good preparation for learning C. 

Arithmetic Operations
Here's an example of an awk program that uses simple arithmetic.

$3 > 10 { print $1, $2, $3-10 }

In the print statement,
$3-10

has the value of the third field in the record, minus 10. This is the value that 
print displays. If you apply this program to the hobbies file (shown in 
Chapter 2: “Basic Concepts”), the output is

Jim reading 5
Linda bridge 2
Katie jogging 4
Andrew wind-surfing 10
Lori weight-lifting 2

You could say that the program worked like this: if someone spends more 
than 10 hours on a hobby, the program displays the person's name, the name 
of the hobby, and how many extra hours the person spends on the hobby 
(that is, the number of hours more than 10).

Something like
$3-10

is called an arithmetic expression. It performs an arithmetic operation and 
comes up with a result; the result of the arithmetic operation is called the 
value of the expression.

awk recognizes the following arithmetic operations

The remainder operation is also known as the modulus or integer remainder 
operation. The value of this expression is the integer remainder you get when 
you divide A by B. For example,

7 % 3

has a value of 1, because when you divide 7 by 3, you get a result of 2 and a 
remainder of 1.

The value for the exponentiation operation
A ^ B

is the value of A raised to the exponent B. For example,
3 ^ 2

has the value 9 (that is, 3*3).

Here are some programs that perform simple arithmetic with the hobbies 
file. Try to figure out what they do and what they display.

(i) $1 == "Katie" { print $2, $3/7 }
(ii) { print $1, $2, $3/7 }
(iii) $1 == "Jim" { print $1, $2, "$", $4/52 }
(iv) print $1, "$", $4*1.05 }

After you have thought about the programs, run them to see if they produce 
the output you have predicted. Here is our interpretation of the programs

(i) Since field three gives the average number of hours per week that a 
person spends on a hobby, $3/7 shows the average number of hours 

Operation Operator Example

Addition A + B 2+3 is 5

Subtraction A - B 7-3 is 4

Multiplication A * B 2*4 is 8

Division A / B 6/3 is 2

Negation - A - 9 is -9

Remainder A % B 7%3 is 1

Exponentiation A ^ B 3^2 is 9

per day. Program (i) therefore displays how many hours per day 
Katie spends on each of her hobbies.

(ii) This is a variation on program (i). It displays how many hours per 
day each person spends on each hobby.

(iii) Field four gives the amount of money a person spent this year on a 
particular hobby. Dividing this by 52 gives the average amount of 
money spent per week. 

(iv) If the current inflation rate is 5 per cent, multiplying this year's 
expenses by 1.05 gives the amount of money the same person might 
expect to spend next year. This is the information that program (iv) 
displays.

Operation Ordering
Expressions can contain several operations, as in

A+B*C

As is customary in mathematics, all multiplications and divisions and 
remainder operations are performed before additions and subtractions. When 
handling this expression, awk performs B*C first and then adds A. The value 
of

2+3*4

is therefore 14 (3*4 first, then add 2). If you want a particular operation done 
first, enclose it in parentheses, as in

(A+B)*C

When evaluating this expression, awk performs the addition before the 
multiplication; therefore

(2+3)*4

is 20 (2+3 first, then multiply by 4).

As an example of this, consider the program
{ print $4/($3*52) }

$4 is the amount of money a person spent on a hobby in the last year. $3 is 
the average number of hours a week the person spent on that hobby, so 
$3*52 is the number of hours in 52 weeks (that is, one year). $4/($3*52) 
is therefore the amount of money that the person spent on the hobby per 
hour.

A table providing the order for awk expressions is provided in the awk 
reference page.

Formatted Output
The output of the program

$1 == "Jim" { print "$", $4/52 }

is
$ 1.92308
$ 0.192308
$ 1.34615

This output shows the amount of money per week that Jim spent on his 
hobbies; however, money amounts usually have only two digits after the 
decimal point. How can you change the program to make the money 
amounts appear more normal? The answer is to use the printf action 
instead of print. This lets you specify the format in which awk displays the 
output. 

A printf action looks like this:
{ printf format-string, value, value, ... }

The format-string indicates the output format. The given values give the data 
to be displayed.

A format string contains two kinds of items:

 Normal characters, which are just displayed as is; and

 Placeholders, which are replaced with values given later in the printf 
action.

As an example, try running the following program on the hobbies file.
$2 == "bridge" { printf "%5s plays bridge\n", $1 }

awk displays:
  Jim plays bridge
Linda plays bridge
 Lori plays bridge

The format string ends in \n; for 
more information, see “Escape 
Sequences” on page 18. 

The format string
"%5s plays bridge\n"

has one placeholder: %5s. When printf displays its output, replacing the 
placeholder with the value $1 which is the first field of the record being 
examined. The rest of the format string is displayed as is. 

Placeholders The form of the placeholder %5s tells awk how to display the associated 
value. All placeholders begin with % and end in a letter. These are some of 
the most common letters used in placeholders:

  c if the associated value is an integer, printf displays the character 
in the native character set with that integer value; if the value is a 
string, printf displays the first character of the string.

  d an integer in ‘decimal’ form (base 10)

  e a floating point number in scientific notation, as in, for example, -
d.ddddddE+dd.

  f a floating point number in conventional form, as in, for example, -
ddd.dddddd.

  g a floating point number in either e or f form, whichever is shorter; 
also, non-significant zeroes are not displayed.

  o an unsigned integer in octal form (base 8).

  s a string

  x an unsigned integer in hexadecimal form (base 16).

For example, the format string
"%s %d\n"

contains two placeholders: %s represents a string and %d represents a 
decimal integer.

Between the % and the letter at the end of the placeholder, you can put 
additional information. If you put an integer, as in %5s, the number is used as 
a width. awk displays the corresponding value using (at least) the given 
number of characters; therefore in

$2 == "bridge" { printf "%5s plays bridge\n", $1 }

the value of the string $1 replaces the placeholder %5s and is always 
displayed using five characters. The output is therefore

  Jim plays bridge
Linda plays bridge
 Lori plays bridge

as shown before. If you just write
$2 == "bridge" { printf "%s plays bridge\n", $1 }

without the 5, the output is:
Jim plays bridge
Linda plays bridge
Lori plays bridge

If no width is given, awk displays values using the smallest number of 
characters possible.

awk also lets you put a minus sign (-) in front of the number in the width 
position. The amount of output space is the same, but the information is left-
justified. For example,

$2 == "bridge" { printf "%-5s plays bridge\n", $1 }

displays:
Jim   plays bridge
Linda plays bridge
Lori  plays bridge

A placeholder for a floating point number may also contain a precision. You 
can write this as a dot (decimal point) followed by an integer. Specifying a 
precision tells printf how many digits to display after the decimal point in 
a floating point number. For example, in

$1 == "John" { printf "$%.2f on %s\n", $4 * 1.05, $2 }

the placeholder %.2f indicates that printf is to display all floating point 
numbers with two digits after the decimal point.

The output of this program is
$105.00 on role-playing
$31.50 on jogging

For good-looking output, you might specify both a width and a precision. 
For example, the program

$1 == "John" {printf "$%6.2f on %s\n", $4 * 1.05, $2}

displays the following:
$105.00 on role-playing
$ 31.50 on jogging

%6.2f indicates that the corresponding floating point value is to be 
displayed with a width of six characters, with two characters after the 
decimal point. 

Here are a few more awk programs that work on the hobbies file. Predict 
what each displays and run them to see if your prediction is right. 

(i) { printf "%6s %s\n", $1, $2 }
(ii) { printf "%20s: %2d hours/week\n", $2, $3 }
(iii) $1=="Katie" { printf "%20s: $%6.2f\n",$2,$4 }

Escape 
Sequences

All of the format strings so far have ended in \n. This kind of construct is 
called an escape sequence. All escape sequences are made from a backslash 
character (\) followed by one to three other characters.

Escape sequences are used inside all strings, not just those for printf, to 
represent special characters. In particular, the \n escape sequence represents 
the newline character. A \n in a printf format string tells awk to start 
displaying output at the beginning of a new line. In a program like

$1 == "Lori" { printf " %s", $2 }

the output is

 jogging weight-lifting bridge
The output is all on one line; without the \n escape sequence, printf does 
not start new lines.

You can use the \n escape sequence in the middle of a format string. For 
example, the output of

$1 == "John" { printf "%s:\n   %d\n",$2,$3 }

is
role-playing:
   8
jogging:
   8

The first newline escape sequence starts a new line after the colon; the 
second starts a new line after the value of $3.

The following table lists escape sequences that may be used in awk strings:

Escape ASCII Character

\a audible bell

\b backspace

\f formfeed

\n newline

\r carriage return

\t horizontal tab

\v vertical tab

\ooo ASCII character, octal ooo

\xdd hexadecimal character dd

\” quote

\c any other character c

The escape sequence \" (a backslash followed by a quote) is used when you 
want a string to contain an actual quote character, as in

"He said, \"Hello\"."

By entering \", you indicate that the quote character is inside the string; if 
you left out the backslash, awk interprets the quote as the end of the string. 

Because a backslash followed by another character looks like an escape 
sequence, you must type two backslashes (\\) if you want to put a single 
backslash character in a string. 

For example,
{ print "The backslash (\\) character" }

displays
The backslash (\) character

Variables
Suppose you want to find out how many people have jogging as a hobby. To 
do this, you have to look through the hobbies file, record-by-record, and 
count the number of records that have jogging in their second field. This 
means that you must remember the count from one record to the next.

awk programs remember information by using variables. A variable is a 
storage place for information. Every variable has a name and a value. An 
awk action of the form

name = value

assigns the specified value to the variable that has the given name. For 
example,

count = 0

assigns the value 0 to the variable count. 

Note Do not confuse = (assigns) with == (is equal to). = stores a value in a 
variable; == tests to see if two values are equal.

You can use variables in expressions. For example, the value of the 
expression

count + 1

is the current value of count, plus 1.

Now consider the action
count = count + 1

awk first finds the value of
count + 1

and then assigns this value to count. Thus the preceding action increases the 
value of count by 1. In a program, you can use this sort of action to count 
how many people have jogging as a hobby.

BEGIN { count = 0 }
$2 == "jogging" { count = count + 1 }
END { printf "%d people like jogging.\n", count } 

Let's look at this program line-by-line.
BEGIN { count = 0 }

BEGIN is a special type of pattern. When a rule has BEGIN as its pattern, awk 
performs the associated action before looking at any of the records in the 
data file. In this example, awk begins by assigning the value 0 to count.

$2 == "jogging" { count = count + 1 }

adds one to count every time awk finds a record with jogging in the second 
field.

END { printf "%d people like jogging.\n", count }

is another rule with a special pattern. When a rule has END as its pattern, awk 
performs the associated action after awk looks at all records in the data files 
specified on the command line. Thus when awk has looked at all the records, 
the printf action displays the count of people who jog. The output from the 
program is

3 people like jogging.

Notice how the value of count was displayed in place of the %d placeholder.

Here are a few more programs that use variables. Examine the programs and 
predict what they do.

(i) BEGIN { count = 0 }
$1 == "John" { count = count + 1 }
END { printf "John has %d hobbies.\n", count } 

Here are the explanations of what the programs do.

With variables, you can write even more complex programs. For example, 
consider the following

BEGIN  { sum = 0; count = 0 }
$2 == "role-playing" {
    count = count + 1
    sum = sum + $4
}
END  {

printf "Average per person: $%6.2f\n",sum/count
} 

This program has two variables. count keeps track of the number of people 
with role-playing as a hobby. sum keeps track of the amount of money spent 
on role-playing. Dividing sum by count, gives the average amount spent on 
role-playing.

Notice that the action part of the BEGIN rule contains two assignment 
instructions. A semicolon is used to separate the two instructions. The 
second rule in the program also has two assignments: 

count = count + 1
sum = sum + $4

These two instructions were put on separate lines. When an action contains 
more than one instruction, you can separate the instructions with semicolons 
or by putting them on separate lines. 

(ii) BEGIN { sum = 0 }
$1 == "Linda" { sum = sum + $4 }
END { 
    printf "Linda spends $%6.2f a year\n",sum
}

(iii) BEGIN  { hours = 0 }
$1 == "Lori" { hours = hours + $3 }
END {
    printf "Lori passes %d hours/week\n",hours
}

(i) This counts the number of hobbies that John has.

(ii) This adds up the amount of money that Linda spent on hobbies 
in the past year.

(iii) This calculates how many hours a week that Lori spends on her 
hobbies.

You can use variables in the pattern part of a rule. For example,
BEGIN  { max = 0 }
$3 > max { max = $3 }
END  {

printf "The maximum time was %d hours.\n", max 
}

finds the maximum value of field three in the hobbies file. The maximum 
is set to 0 to start. Then, if a record has a greater value in field three, max is 
set to this new value. At the end of the data file, max holds the largest value 
found.

As an exercise, you should try to write an awk program that examines the 
hobbies file and calculates the average number of hours per week that 
someone spends on any one hobby. After that, write a program that 
calculates the average number of hours per year that a person spends on any 
one hobby.

The Increment 
and Decrement 
Operators

The last section showed how to advance the value held in a variable, with
count = count + 1

This is such a common operation that awk has a special operator for 
incrementing variables by 1.

count++

adds 1 to the current value of count.

The -- operator is the counterpart of ++. It decrements (subtracts 1 from) the 
current value of a variable. For example, to subtract 1 from count, you would 
write: 

count--

Initial Values If you use any variable X in an arithmetic expression before you assign the 
variable a value, X is automatically given the value 0. This means that in the 
program 

BEGIN  { count = 0 }
$2 == "jogging" { count = count + 1 }
END  { printf "%d people jog\n", count }

you could omit the BEGIN rule. You do not have to assign 0 to count 
explicitly; count automatically has the value 0 the first time you use it.

Built-In 
Variables

awk has a number of built-in variables that you can use in your programs. 
You do not have to assign values to these variables —awk automatically 
assigns the values for you. The following list describes some of the 
important numeric built-in variables.

NR contains the number of records read so far. When awk is looking at 
the first record, NR is 1; when awk is looking at the second record, 
NR is 2; and so on. In a BEGIN rule, NR has the value 0. In an END 
rule, NR contains the total number of records read. A rule like

END  { print NR }

displays the total number of data records read by the awk program.

FNR is like NR, but it counts the number of records that have been read 
so far from the current file. When you give several data files on the 
awk command line, awk sets FNR back to 1 when it begins reading 
each new file. 

Thus a command like 
{ printf "%d:%s\n",FNR,$0 }

displays the line number in the current file, followed by a colon, 
followed by the contents of the current line.

NF gives the number of fields in the current record. For the hobbies 
file, NF is 4 for each line, because there are four fields in each 
record. In an arbitrary text file, NF gives the number of words on the 
current line in the file; by default, awk assumes that blanks separate 
the fields of a record, so it considers each word on a line to be a 
separate field; therefore, the program

{ count = count + NF }
END  { print count }

displays the total number of words in the file.

You can use built-in variables in place of any other variable or value. For 
example, they may appear in the pattern part of a rule.

NF > 10 { print }

displays any record that has more than ten fields.
NR == 5 { print }

displays record 5 in a file —the pattern selection criterion is true only when 
NR is 5.

As another example, try to predict what
{ print $NF }

does. Since NF is the number of fields in the current record, it is also the 
number of the last field in the record; therefore $NF refers to the contents of 
the last field in a record and 

{ print $NF }

displays the last field in every record in the data file.

To test your understanding of almost everything discussed in this chapter, try 
to predict what

(NR % 5) == 0

displays. The expression
NR % 5

calculates the remainder of NR divided by 5. The rule displays a record 
whenever this remainder is equal to 0; therefore, the rule displays every fifth 
record from the data file. 

As an exercise, write awk programs to do the following.

Go ahead and write these programs now. Test them by running them on 
arbitrary text files. Once you have a solution that works, compare them 
against the following possible answers.

(i) NF != 3

(ii)   { words = words + NF )
END  {

printf "Words = %d, Lines = %d\n", words, NR
}

(iii) NF == 4  { count = count + 1 }
NF == 5  { count = count + 1 }
END  { print count }

(iv) { words = words + NF }
END  { printf "Average = %d\n", words/NR }

(i) Display every record which does not have exactly 3 fields.
(ii) Display the total number of words and total number of lines 

in a text file. (This is two thirds of what the wc command 
does.)

(iii) Display the total number of records that have either four 
fields or five fields.

(iv) Display the average number of words per line in a text file.

Arithmetic Functions
In awk, a function can be compared to a car assembly line: you feed in 
various parts and raw materials at one end, and you get out a complete 
product at the other end. Of course, in awk a function is fed data values 
(called the arguments of the function) and the final product is also a data 
value (called the result of the function). 

You may already be familiar with this kind of function in mathematics. For 
example, sin() is a mathematical function. If you feed an angle into the 
sin() function, you get out a number which is the trigonometric sine of the 
given angle. The angle is the argument of the function and the sine is the 
result.

In awk, you use functions inside expressions. For example, you might have 
the assignment

y = sin(x)

The right hand side of the assignment is a function call. The name of the 
function is sin; this name is immediately followed by parentheses which 
enclose the arguments of the function. When an awk program contains a 
function call, awk calculates the result of the function, then use that result in 
the expression that contains the function call. In the preceding assignment, 
awk calculates the number that is the sine of the given angle, and then 
assigns that number to the variable y.

As another example, sqrt() is an awk function the result of which is the 
square root of its argument.

The assignment
x = sqrt(16)

assigns the value 4 to x.

To show how you can use these functions, consider a set of data that contains 
one number per line. Here's a program which reads this number and displays 
the square root. 

{ printf "Number: %f, Root: %f\n", $1, sqrt($1) }

You might run this program with the command line
awk '{printf "Number: %f, Root: %f\n", $1, sqrt($1)}'

and then type in numbers from the terminal. Each time you press ENTER at 
the end of the line, awk displays the square root of the number. 

Any argument of a function may be an expression instead of a single value. 
For example,

y = sin(2*x)

awk calculates the value of the expression, then uses that value as the 
argument of the function.

awk recognizes the most common mathematical functions, as shown in the 
accompanying table.

Several of these functions may require more explanation.

The int() function takes a floating point number as an argument and 
returns an integer. The integer is just the floating point number, without its 
fractional part. For example,

int(6.3)

has the value 6, while
int(-7.4)

has the value -7. Notice that the fractional part is just removed, not rounded.
int(8.99999)

has the value 8.

Every call to rand() returns a new random number between 0 and 1. In this 
way, you can get a sequence of random numbers. You can use srand() to 
set the starting point or seed for a random number sequence. If you set the 
seed to a particular value, you always get the same sequence of numbers 
from rand(). This is useful if you want a program to use rand() but obtain 
uniform results every time the program runs. 

Function Result

sqrt(x) square root of x

sin(x) sine of x, where x is in radians

cos(x) cosine of x, where x is in radians

atan2(y,x) arctangent of y/x in range - to 

log(x) natural logarithm of x

exp(x) the constant e to the power x

int(x) integer part of x

rand() random number between 0 and 1

srand(x) sets x as seed for rand()

As an example of how rand() can be used, here's a sequence of instructions 
that you could use in an awk program to simulate a roll of two six-sided dice. 

die1 = int( 6 * rand() + 1 )
die2 = int( 6 * rand() + 1 )

The function call
rand()

obtains a random floating point number between 0 and 1 (not including 1). 
Notice that the function call needs the parentheses, even though rand() 
doesn't need any argument values. Multiplying the random number by 6 
gives a floating point value between 0 and 6 (not including 6). Adding 1 
gives a floating point value between 1 and 7 (not including 7). applying the 
int() function to this floating point value drops the fraction part, giving us 
an integer between 1 and 6.

4Patterns and Regular 
Expressions

So far, this document has discussed three kinds of patterns: comparisons, 
and the special patterns BEGIN and END. This chapter discusses a fourth 
kind: regular expressions. 

A regular expression is a way of telling awk to select records that contain 
certain strings of characters. For example,

/ri/ { print }

tells awk to display all records that contain the string ri. If you apply this rule 
to the hobbies file, you get

Jim     bridge            4      10.00
Linda   bridge           12      30.00
Lori    jogging           5      30.00
Lori    weight-lifting   12     200.00
Lori    bridge            2       0.00

All these records contain ri, either in Lori or bridge.

Regular expressions are always enclosed in slashes as shown in the rule just 
discussed. For example, you might write

/ing/

to display all the records that contain ing.

awk pays attention to the case of letters in regular expressions. For example,
/li/

displays the record that contains weight-lifting; however, the /li/ does 
not match the Linda records because the L in Linda is uppercase. 

It is important to recognize the difference between two rules like
$1 == "Lori"
/Lori/

To satisfy the pattern
$1 == "Lori"

a record must have its first field exactly equal to the string Lori. If the first 
field is Lorie, for example, the comparison is not true and the pattern is not 
satisfied. With the regular expression

/Lori/

the string Lori can appear anywhere in the record, and can be all or part of a 
field. This regular expression would match a string like Lorie. 

The Matching Operators
If the pattern in a rule is just a regular expression, awk looks for a matching 
string anywhere in a record. Sometimes however, you only want to look for a 
matching string in a particular field of a record. In this case, you can use a 
matching expression.

There are two types of matching expressions:
string ~ /regular-expression/

is true if the string matches the given regular expression. (The ~ character is 
called tilde.)

string !~ /regular-expression/

is true if the string does not match the given regular expression. For 
example,

$2 ~ /ri/

displays all records that have ri somewhere in the second field. The output is:
Jim     bridge            4      10.00
Linda   bridge           12      30.00
Lori    bridge            2       0.00

The rule
$1 !~ /J/

displays all records without a J somewhere in the first field. These two 
patterns are equivalent:

/Lori/
$0 ~ /Lori/

Metacharacters
The following characters have special meanings when you use them in 
regular expressions.

^ stands for the beginning of a field. For example,
$2 ~ /^b/ { print }

displays any record with a second field that begins with b.

$ stands for the end of a field. For example,
$2 ~ /g$/ { print }

displays any record with a second field that ends with g.

. matches any single character (except the newline). For 
example,

$2 ~ /i.g/ { print }

selects records with fields containing ing, and also 
records containing bridge (idg).

| means or. For example,
/Linda|Lori/

is a regular expression that matches either of the strings 
Linda or Lori.

* indicates zero or more repetitions of a character. For 
example,

/ab*c/

matches abc, abbc, abbbc, and so on. It also matches ac 
(zero repetitions of b). The * is most frequently used in the 
expression .*. Since . matches any character except the 
newline, .* matches an arbitrary string of zero or more 
characters. For example,

$2 ~ /^r.*g$/ { print }

displays any record with a second field that begins with r, 
ends in g, and has any set of characters between (for 
example, reading and role-playing).

+ is similar to *, but stands for one or more repetitions of a 
string. For example,

/ab+c/

matches abc, abbc, and so on, but does not match ac.

\{m,n\} indicates m to n repetitions of a character (where m and n 
are both integers). For example,

/ab\{2,4\}c/

would match abbc, abbbc, and abbbbc, and nothing else.

? similar to *, but stands for zero or one repetitions of a 
string. For example,

/ab?c/

matches ac and abc, but not abbc, etc.

[X] matches any one of the set of characters X given inside the 
square brackets. For example,

$1 ~ /^[LJ]/ { print }

displays any record with a first field that begins with either 
L or J. awk also offers some special cases for common sets 
of characters. 

[:lower:] stands for any lowercase letter

[:upper:]stands for any uppercase letter

[:alpha:]stands for any letter

[:digit:]stands for any digit

[:alnum:]stands for any alphanumeric character (letter 
or digit)

[:space:]stands for any white space character (such as a 
space or a tab character)

[:print:]stands for any printable character, including 
the space character

[:graph:]stands for any printable character except the 
space character

[:punct:]stands for any printable character that isn't 
white space or alphanumeric

[:cntrl:]stands for any non-printable character

Thus
/[[:digit:][:alpha:]]/

matches a digit or letter. If you use these special cases, you 
don't have to remember, for instance, that
/[,\./?\[\]{}=+_-~`!@#\$%:;"'\^&\*()|\\]/

matches any printable character that isn't white space, 

numbers, or letters—you can use /[[:punct:]]/ 
instead.

[^X] matches any one character that is not in the set X that 
follows the caret ^. For example,

$1 ~ /^[^LJ]/ { print }

displays any record with a first field that does not begin 
with L or J.

$1 ~ /^[^[:digit:]]/ { print }

displays any record with a first field that does not begin 
with a digit.

(X) matches anything that the regular expression X does. You 
can use parentheses to control the way in which other 
special characters behave. For example, * normally 
applies to the single character that immediately precedes 
it. This means that

/abc*d/

matches abd, abcd, abccd, and so on; however,
/a(bc)*d/

matches ad, abcd, abcbcd, abcbcbcd, and so on.

The characters with special meanings are
^   $   .   *   +   ?   [   ]   (   )   |

These are known as metacharacters. When a metacharacter appears in a 
regular expression, it usually has its special meaning. If you want to use one 
of these characters literally (without its special meaning), put a backslash in 
front of the character. For example,

/\$1/ { print }

displays all records that contain a dollar sign $ followed by a 1. If you just 
wrote

/$1/ { print }

awk would search for records where the end of the record was followed by a 
1 - which is impossible.

Because the backslash has this special meaning, \ is also considered a 
metacharacter. If you want to create a regular expression that matches a 
backslash, you must therefore use two backslashes (\\).

Matching Expressions with Strings
Previous sections have shown matching operations that contain regular 
expressions inside slash (/) characters. Matching operations can also refer to 
normal strings, as in: 

$1 ~ "xyz"

This has the same effect as
$1 ~ /xyz/

awk compiles regular expressions when it reads the program. To use a string 
as a regular expression, awk constructs a dynamic regular expression out of 
the string. This is slower, since the string is compiled into a dynamic regular 
expression every time it is used; however, it is much more powerful.

When a matching operation uses a string instead of a regular expression, and 
the string contains one or more metacharacters, the situation is a little bit 
tricky. If you want to escape a metacharacter (have it taken literally), you 
must use two backslashes instead of one. For example, suppose you want to 
look for strings of the form "$1.00" in field 4 of a record. Using regular 
expressions, you would write

$4 ~ /\$1\.00/

to show that both the $ and the . should be taken literally. With strings, you 
would have to write:

$4 ~ "\\$1\\.00"

You need two backslashes instead of one. The reason is simple: as discussed 
in Chapter 3: “Simple Arithmetic”, you need to type two backslashes inside 
a quoted string to get the effect of one. For example,

{ print "The backslash character: \\" }

displays

The backslash character: \
To match a backslash in a dynamic regular expression, you must use four, as 
in

$1 ~ "\\\\"

awk reads the literal string "\\\\" and turns it into a string consisting of 
"\\". When used as a dynamic regular expression, this matches one 
backslash. 



Pattern Ranges

Using MKS AWK 35

Pattern Ranges
A rule of the form

pattern1 , pattern2 { action }

performs the given action on every line, starting at an occurrence of pattern1 
and ending at the next occurrence of pattern2 (inclusive). For example, the 
rule

NR == 1, NR == 10 { print $1 }

displays the first field of each of the first 10 input lines. It starts when NR is 
1 and ends when NR is 10.

/Jim/, /Linda/ { print $2 }

displays the second field of all lines between an occurrence of Jim and an 
occurrence of Linda.

Using the hobbies file as our data file, the output is
reading
bridge
role-playing
bridge

When awk finds a record matching pattern2, it begins to look for a line 
matching pattern1 again. Thus, with a rule like 

/reading/, /role/

the output is
Jim     reading          15     100.00
Jim     bridge            4      10.00
Jim     role-playing      5      70.00
Katie   reading          10      60.00
John    role-playing      8     100.00

awk displays the first range of records from reading to role, then starts 
looking for reading again.

It's important to bear in mind that awk starts performing the rule's action as 
soon as there is a record that matches pattern1. awk does not check to make 
sure that there is a line matching pattern2 in the rest of the file. This means 
that

/Lori/, /Jim/ { print $2 }

begins displaying at the first record that contains Lori, and keeps going 
until it reaches the end of the file —no Jim is found.

Multiple Conditions
The && (double ampersand) operator means AND. For example,

$3 > 10 && $4 > 100.00 { print $1, $2 }

displays the first and second fields of any record where $3 is greater than 10 
and $4 is greater than 100.00. Similarly, the rule 

$1 ~ /J/ && $4 < 50.00

displays all records in which $1 contains a J and $4 is less than 50.00.

The || (double or-bar) operator means OR. For example,
$1 == "Linda" || $1 == "Lori"

displays any record with a first field that is either Linda or Lori.
/jogging/ || /reading/ { sum = sum + $4 }
END  { print sum }

calculates the total spent by the hobbyists on both jogging and reading (since 
sum is increased if the hobby is either jogging or reading). The 
preceding program is equivalent to

/jogging|reading/ { sum = sum + $4 }
END  { print sum }

which demonstrates that there are often several ways of writing the same 
program.

You can only use the && and || operators to combine complete pattern 
expressions. For example, you cannot write

$1 == "Linda" || "Lori"

You must write
$1 == "Linda" || $1 == "Lori"

For practice with the concepts discussed in this chapter, write programs that 
do the following.

(i) Display every record that begins with A and contains more than 
four fields.

(ii) Display the number of records that contain a dollar sign ($).

(iii) Display records 10 through 20 of every data file read in.

(iv) Display every 10th record of a file, plus the record that 
immediately follows the 10th record (that is, records 10 and 11, 
records 20 and 21, and so on).

When you have written your programs, compare them against the solutions 
that follow. Remember that there may be several ways to write the same 
program.

(i) /^A/ && NF > 4

(ii) /\$/ { count = count + 1}
END { print count }

(iii) FNR == 10, FNR == 20

(iv) (NR % 10) == 0, (NR % 10) == 1
or
((NR % 10) == 0) || ((NR % 10) ==1)

5Actions and Control 
Structures

Previous chapters have discussed three actions: print, printf, and 
assignments. This chapter examines a wide variety of constructs that may 
appear in the action part of an awk rule. Note that most of these are virtually 
identical to constructs in the C programming language.

Comments
A comment is a note inside your program, explaining what the program is 
doing. awk skips over comments, ignoring them; in this way, comments do 
not affect how your program behaves, but they should help explain what's 
going on.

A comment begins with a # character. When awk sees the # in a program 
(outside of a quoted string or regular expression), awk ignores the rest of the 
line. For example, you might write

# This program totals the hours John spends on hobbies
/John/ { sum = sum + $3 }    # field 3 is hours
END { print sum } 

The first line of the program explains what the program is doing. This is 
useful when you have a number of awk programs stored in different files and 
you cannot remember which program is which. A simple comment at the 
beginning of the program lets you identify the program without having to 
read through the code and figure out what is going on. The line

/John/ { sum = sum + $3 }       # field 3 is hours

shows another way in which you can use comments. A comment on the end 
of a line can give further information about what that line is doing. In this 
case, it explains the meaning of the number in field three of the record.

Use comments in any significant program. Without meaningful comments, 
you may find it very difficult to understand a program if you look at it 
several months after you wrote it. Comments also make it easier for others to 
understand the programs you write.

The if Statement
An if statement is an action of the form

if (expression) statement1 else statement2

Typically, the expression in the if statement has a true/false value. If the 
value is true, statement1 is performed; otherwise, statement2 is performed. 
The else statement2 part is optional. 

Note When the else statement2 part of the if statement is on the same line as 
the statement1, you must separate the else statement2 from statment1 with a 
semicolon (;).

As an example of how to use if, you are going to write some programs that 
examine a file of baseball scores (since you are probably tired of the hobbies 
example by now). The baseball file is installed as ROOTDIR/samples/
guide/baseball, and it looks like this:

Blue Jays       8     Yankees        6
Brewers         3     Red Sox        7
Tigers         10     Orioles        3
        ...and so on... 

Each line gives the home team first and the visitors second. Fields in each 
record are separated by tab characters (shown here as wide spaces) instead of 
single blanks, because some team names contain blanks. This means that 
you must specify the option

-F"\t"

in all awk programs that you run on the baseball file. This option is 
equivalent to having the line:

BEGIN   { FS = "\t" }

in your awk program.

All the scores in the baseball 
file were generated randomly 
with an awk program shown in 
Chapter 8: “User-Defined 
Functions”.

The program
{

if ($2 > $4) print "Home"
else print "Visitor"

}

displays Home when the home team's score ($2) is greater than the visiting 
team's, and displays Visitor otherwise. (For those who are not baseball 
fans, it should be pointed out that baseball games do not end in ties —they 
keep playing until one team wins.

You may omit the else part of an if statement. In this case, awk does nothing 
if the expression of the if statement is not true. For example,

$1 ~ /Tigers/  { if ($2 > $4) win++ }
END  { print win }

is a simple program that looks at all the Tigers' home games and displays the 
number of times the Tigers won. On records where $2 is not greater than $4, 
awk takes no action.

As a more complicated example, consider this:
$1 ~ /Yankees/ { if ($2 > $4) print "Home Win"

else print "Home Loss" }
$3 ~ /Yankees/ { if ($4 > $2) print "Away Win"

else print "Away Loss" }

This program runs through the baseball scores looking for games involving 
the Yankees, displaying appropriate messages for each possible outcome.

$1 ~ /Yankees/ {
                   if ($2 > $4) hw++
                   else         hl++
               }
$3 ~ /Yankees/ {
                   if ($4 > $2) aw++
                   else         al++
               }
END  {
         printf "Home Wins: %d\n", hw
         printf "Home Losses: %d\n", hl
         printf "Away Wins: %d\n", aw
         printf "Away Losses: %d\n", al
     }

is similar to the previous program; however, this example keeps track of the 
number of wins and losses, at home and away, then displays these values at 
the end. 

A Word on Style
Take note of the way in which indentation was used in the preceding 
program.

 Except in trivial cases, a new line starts after every opening brace ({).

 Every else is lined up under the corresponding if.

 Parallel statements, like the sequence of printf instructions are lined 
up underneath each other. 

It is not necessary to write awk programs in this way, but appropriate 
indentation and spacing make programs much easier to read and understand. 
Your style for writing programs can also help you spot errors as you type in 
your program. For example, if you always try to make opening and closing 
braces line up, it is easy to notice when you have left out one of the braces. 

The indentation format used in this document is just one example of a clean 
readable programming style. All programmers develop personal preferences 
as they become familiar with a language, and you may decide to deviate 
from this guide's style in some respects. The most important thing is to have 
a style and to follow it consistently in all your future programs. It may not 
make much difference now, when your programs are relatively simple; but 
as your programs become more complex, you will find that style is an 
important aid to writing programs that work correctly. 

Compound Statements
In an if statement, you sometimes want to perform several instructions in 
the if or the else part. You can do this by enclosing the instructions in 
brace brackets. Such a construct is called a compound statement.

For example, consider the following program.
{
    if ($2 > $4) {
        homewin++
        printf "The %s defeated the %s.\n", $1, $3
    } else {
        homeloss++
        printf "The %s defeated the %s.\n", $3, $1
    }
}
END  {
    printf "The home team won %d times.\n", homewin
    printf "The home team lost %d times.\n", homeloss
}

awk applies the first action to every record in the file. It keeps a count of how 
many times the home team wins and how many times the home team loses. 
It also displays a line telling who defeated whom. The END action 
summarizes the results after they have been calculated.

As another example, the following program examines the games involving 
the Orioles.

$1 ~ /Orioles/ {
    if ($2 > $4) {
        win++    # Home win
        printf "%s: %d, %s: %d\n",$1,$2,$3,$4
    } else {
        loss++    # Home loss
        printf "%s: %d, %s: %d\n",$3,$4,$1,$2
    }
}
$3 ~ /Orioles/ {
    if ($4 > $2) {
        win++    # Away win
        printf "%s: %d, %s: %d\n",$3,$4,$1,$2
    } else {
        loss++     # Away loss
        printf "%s: %d, %s: %d\n",$1,$2,$3,$4
    }
}
END  {
    printf "Wins: %d, Losses: %d\n", win, loss

}

Each line of output from the first two actions has the form
Winning team: score, Losing team: score

The final line of output (from the END rule) summarizes the Orioles' wins 
and losses.

Examine this program closely to see how it works. The program is quite 
straightforward, but make sure you understand how it covers all the possible 
cases. 

One if statement can contain another. For example, the previous program 
can be written

/Orioles/ {
    if ($2 > $4) {    # Home team wins
        printf "%s: %d, %s: %d\n",$1,$2,$3,$4
        if ($1 ~ /Orioles/)
            win++
        else
            loss++
    } else {          # Home team loses
        printf "%s: %d, %s: %d\n",$3,$4,$1,$2
        if ($3 ~ /Orioles/)
            win++
        else
            loss++
    }
}
END  {
    printf "Wins: %d, Losses: %d\n", win, loss
}

This version of the program determines whether the game was won by the 
home team or not, displays the scores with the winner first, and then checks 
to see if the Orioles were the home team or the visitors. The previous version 
of the program split the problem into two parts: one action performed when 
the Orioles were the home team and one when they were not.

while Loops
A while loop repeats one or more other instructions as long as a given 
condition holds true. The format of the loop is

while (expression) statement

where the statement can be a single statement or a compound statement. For 
example, the file numbers contains a set of one to ten random numbers on 
each line. The following program adds up the numbers on each line and 
displays the line's total.

{
    sum = 0
    i = 1
    while (i <= NF) {
        sum = sum + $i
        i = i + 1
    }
    print sum
}

The variable i counts fields in the record. While i is less than or equal to the 
total number of fields in the record, the while loop adds the value of the ith 
field to sum and then adds 1 to i. The loop then starts again; if the new value 
of i is still less than or equal to the total number of fields, the loop adds in 
the value of the next field. The loop stops looping when i is greater than NF.

As another example, here's a program that uses the same data file and 
displays the maximum value on each line.

{
    max = $1        # starting max is field 1
    i = 2
    while (i <= NF) {
        if ($i > max) max = $i
        i = i + 1
    }
    print max
}

On each line, the variable max starts out with the value of the first field (that 
is, the first number). The while loop then moves across the record number 
by number, using an if statement to test whether a field is greater than the 
current value of max. If a greater value is found, max is assigned the new 
maximum value. After the loop, the maximum value is displayed. 

What does the preceding program do if there is only one number on a 
particular line? In this case, NF is 1. awk performs the statements 

max = $1
i = 2
while (i <= NF) ...

and finds that i is already greater than NF; therefore, awk does not perform 
any of the instructions in the while loop at all. If the condition part of a 
while loop is false when the loop is first encountered, the statements in the 
loop are not run at all.

As an exercise, try to write a program that reads a normal text file and writes 
out the text, one word per line.

for Loops
The statement

for (expression1;expression2;expression3) statement

is equivalent to the following instruction sequence.
expression1
while (expression2) {
    statement
    expression3
}

For example,
{
    for (i = NF; i > 0; i--)
        printf "%s ", $i
    printf "\n"
}

goes through a file line-by-line, displaying the words on each line in reverse 
order. The program that displayed the maximum value in an input line can be 
written: 

{
    max = $1
    for (i = 2; i <= NF; i++)
        if ($i > max) max = $i
    print max
}

There is another form of the for 
loop described in Chapter 7: 
“Arrays”.

Therefore, the for loop is just a short-hand way of writing a certain kind of 
while loop.

The next Statement
The instruction

next

skips immediately to the next record in the data file. Try this example on the 
baseball score file:

{
    if (NF < 5) {
        printf "Not enough fields: %s\n", $0
        next
    }
    if ($2 > $4) print "Home Win"
    else print "Home loss"
}

If a particular record has less than five fields, this program displays a 
warning message and skips to processing the next record. This bypasses the 
rest of the instructions in the rule. It also bypasses any other rules that might 
normally be applied to this record. As this example shows, next is often used 
when a program finds a record that does not have the format you expect.

You can also use next to skip to the next record if you do not want the record 
processed by any of the remaining rules. For example, you might write: 

$1 ~ /Orioles/  {count++; next}
$3 ~ /Orioles/  {count++}

This prevents the record from being counted twice if it has Orioles in both 
the first and third fields. Another way to write this program is:

($1 ~ /Orioles/) || ($3 ~ /Orioles/) { count++ }

If you use the next instruction inside a BEGIN rule, it tells awk to start normal 
processing (by reading the first record of the first file). In other words, it 
indicates that you are finished BEGIN processing.

The exit Statement
The statement

exit

makes an awk program behave as if it has just reached the end of data input. 
No further input is read. If there is an END action, awk runs it before the 
program terminates. As with next, exit is often used when input data is 
found to be in error.

If exit appears inside the END action, it terminates the program 
immediately.

6String Manipulation

Preceding chapters have used quoted strings extensively. This chapter 
discusses strings in more detail, and the various operations that manipulate 
strings. 

String Variables
Chapter 3: “Simple Arithmetic” showed how to use numeric variables: 
variables that contained numbers. Variables can also contain strings. For 
example,

a = "string"

assigns a string to a variable a. As an example of how you can use this, here 
is a simple program that checks a text file for duplicate lines (that is, places 
where two adjacent lines are identical).

{
    if ($0 == lastline) printf "%d: %s\n", FNR, $0
    lastline = $0
}

The variable lastline represents the contents of the previous line in the 
file. In the action of the program, awk compares the current record $0 to the 
previous record (stored in lastline). If the two are equal, the printf 
action displays the line number FNR and the contents of the line. At the end 
of the action, lastline is assigned the contents of the current line (so that 
this can be compared to the next line).

You might wonder what lastline contains when the program first begins. 
After all, nothing is assigned to lastline until the first line has been read. All 
string variables begin with a null string value. A null string is a string that 
contains no characters. It is written "". When used in an arithmetic 
expression, a null string has the value 0. 

As another example of using string variables, here's a program that displays 
the last line of a file.

    { line = $0 }
END { print line }

The value of each input line is assigned to the variable line. At the end of the 
file, line contains the contents of the last line in the file; therefore, the END 
action displays the contents of that line.

Built-In String Variables
The “Simple Arithmetic” chapter also discussed the built-in numeric 
variables NF, NR, and FNR. In addition, awk provides a number of built-in 
string variables.

FILENAME contains the name of the current input file. For example, 
when applying programs to the hobbies file, the value of 
FILENAME is hobbies. If the input is coming from the 
awk standard input, the value is -.

FS is the field separator string, giving the character that is 
used to separate fields in the current file. The default value 
for FS is " " (a single blank), which as a special case 
matches both blank and tab; however, if the command line 
contains a -F option specifying a different field separator, 
FS is a string containing the given separator character. As 
well, a program may also assign values to FS to indicate 
new field separator characters. For example, you can 
create a data file with a first line that gives the character 
used to separate fields in the records in the rest of the file. 
An awk program can then use the rule

FNR == 1 { FS = $0 }\

Please see the Input section of 
the awk reference page in the 
PTC MKS Toolkit Utilities 
Reference for details.

This says that the field separator string FS is to be assigned 
the contents of the first record in the current data file. The 
character in this line is then taken to be the field separator 
for the rest of the file (unless FS changes value again). 
Any FS value of more than one character is used as a 
regular expression. 

RS is the input record separator. Just as FS indicates the 
character that separates fields within records, RS indicates 
the character that separates one record from another. By 
default, RS contains a newline character, which means that 

input records are separated by newline characters; 
however, a different character may be assigned to RS; for 
example, with

RS = ";"

input records are separated by semicolon characters. This 
lets you have several records on one line, or a single 
record that extends over several lines—records are 
separated by semicolons, not newlines. As an important 
special case,

RS = ""

separates records by empty lines.

OFS gives the output field separator string. When you use the 
print action to display several values, as in 

{ print A, B, C }

awk displays the output field separator string between 
each of the values. By default, OFS contains a single blank 
character, which is why output values are separated by a 
single blank; however, if you make the assignment

OFS = " : "

the output values are separated by the given string. You 
can also use the OFS to reconstruct the $0 field during 
field assignment.

ORS gives the output record separator. When you use the 
print action to display records, awk displays the output 
record separator at the end of each record. By default, ORS 
is the newline character, which is why print displays a new 
output line each time it is called; however, a different 
separator string may be used by assigning the string to 
ORS.

OFMT is the default output format for numbers when they are 
displayed by print. This is a format string like the one used 
by printf. By default, it is %.6g, indicating that numbers 
are to be displayed with a maximum of six digits after the 
decimal point. By changing OFMT, you can obtain more or 
less displayed precision.

CONVFMT is the default format which awk uses when converting 
numbers to strings internally. This differs from the OFMT 
variable which is only used when displaying numbers. The 
internal conversion of a number to a string occurs when 

you perform catenation, indexing, and some comparison 
operations. awk converts floating point numbers (that is, 
ones that are not integers) to strings as if you had specified 
the operation

sprintf(CONVFMT, number ...)

By default, the value of CONVFMT is %.6g. The next 
session discusses conversions in more detail. 

CONVFMT is a POSIX extension not found in traditional 
implementations of awk.

String vs. Numeric Variables
Earlier sections stated that string variables start out with the null string value, 
while numeric variables start out as 0. There's an obvious question: how can 
awk differentiate between string and numeric variables, especially when it 
has started running the program and you have not even used the variable yet? 
The answer is that a variable is assumed to contain a string unless you use it 
as a number. For example, if you have a program that consists of

{ print X }

with no value assigned to X, awk assumes the variable is a string. Thus the 
output is a blank line for each line of input; if X had been taken as a number, 
the output would be 0 for each line of input.

In an action like
X = $1

the variable X is taken as a number if the form of $1 looks like a number and 
string otherwise. For example, if the record is

3 ...

the first field looks like a number, so X is normally taken to be a numeric 
variable. If the record is

7ABC ...

the first field cannot be a number (even though it starts with a digit), so X is 
taken to be a string variable.

There are times when you want a value to be treated as a string, even though 
it looks like a number. For example, suppose a file contains the string 1e1. 
In some contexts, this could be a number (with an exponential part); in other 

contexts, you might want to interpret this as a string. To make sure that a 
value is taken as a string, even when it might look numeric, concatenate it 
with an empty string, by placing a "" after it, as in

X = $2 ""

This makes sure that awk interprets the value in $2 as a string, even if it 
looks like a number; therefore, X is a string variable. Similarly, if you want to 
make sure that a value is taken to be a number, just add zero to it, as in

X = $3 + 0

In this case, $3 is definitely taken to be a number because it is involved in an 
arithmetic operation. (What happens if $3 is not a valid number? If $3 starts 
with something that looks like a number, as in 7ABC, the numeric value of 
the string is the number. Thus the numeric value of 7ABC is 7. If the field 
does not start with anything that looks like a number, the numeric value of 
the string is zero. Thus the numeric value of ABC is 0.)

String Concatenation
The expression

$2 ""

seen in a previous chapter is an example of string concatenation. When a 
line in a program contains two or more strings that are only separated by 
blank characters, the strings are concatenated (joined) into one long string. 
For example, the action

{ print $1 $2 $3 }

displays the contents of the first three fields, joined together into one string. 
For example, if an input line contains

A B C

the output is
ABC

Similarly, applying the program
$1 ~ /John/ { print "$" $4 }

to the hobbies file, gives this output:
$100.00
$30.00

The $ character is concatenated to the contents of the fourth field in all the 
appropriate records.

String Manipulation Functions
“Simple Arithmetic” on page 13 introduced numeric functions like sin() 
and sqrt(). awk also has a number of functions that perform string 
operations. 

length returns an integer which is the length of the current record 
(that is, the number of characters in the record, without the 
newline on the end). For example, the following program 
calculates the total number of characters in a file (except 
for newline characters):

    { sum = sum + length }
END { print sum }

length(s) returns an integer which is the length of the string s. For 
example, the following program displays the length of the 
first field in each record of the data file:

{ print length($1) }

length($0) is equivalent to just length.

gsub(regexp,replacement)
puts the replacement string replacement in place of every 
string matching the regular expression regexp in the 
current record. For example, the program

{
    gsub(/John/,"Jonathan")
    print
}

checks every record in the data file for the regular 
expression John. Every matching string is replaced with 
Jonathan and displayed. As a result, the output of the 
program is exactly like the input, except that every 
occurrence of John is changed to Jonathan. This form of 
the gsub() function returns an integer which tells how 
many substitutions were made in the current record. This 
is 0 if the record has no strings that match regexp. 

sub(regexp,replacement)
is similar to gsub() except that it only replaces the first 
occurrence of a string matching regexp in the current 
record.

gsub(regexp,replacement,string_var)
puts the replacement string replacement in place of every 
string matching the regular expression regexp in the string 

string_var. For example, the program 
{

   gsub(/John/,"Jonathan",$1)
    print

}

is similar to the previous program, but the replacement is 
only made in the first field of each record. This form of the 
gsub() function returns an integer which tells how many 
substitutions were made in string_var.

sub(regexp,replacement,string_var)
is similar to the previous version of gsub() except that it 
only replaces the first occurrence of a string matching 
regexp in the string string_var. 

As described in the section “Matching Expressions with 
Strings” on page 34, you must use four backslashes to 
embed one literal backslash in a gsub() or sub() 
substitution string. For example,

gsub(/backslash/,"\\\\")

replaces all occurrences of the word backslash with the 
single character \.

index(string,substring)
searches the given string for the appearance of the given 
substring. If the substring cannot be found, index() 
returns 0; otherwise, index() returns the number (origin 
1) of the character in string where substring begins. For 
example,

index("abcd","cd")

returns the integer 3 because cd is found beginning at the 
third character of abcd.

match(string,regexp)
determines if string contains a substring that matches the 
regular expression (pattern) regexp. If so, match() 
returns an index giving the position of the matching 
substring within string; if not, match() returns 0. 
match() also sets the variable RSTART to the index where 
the matching string starts, and the variable RLENGTH to the 
length of the matching string.

substr(string,pos)
returns the last part of string beginning at a particular 
character position. The argument pos is an integer, giving 

the number of a character. Numbering begins at 1. For 
example, the value of

substr("abcd",3)

is the string cd.

substr(string,pos,length)
returns the part of string that begins at the character 
position given by pos and has the length given by length. 
For example, the value of

substr("abcdefg",3,2)

is cd (a string of length 2 beginning at position 3).

sprintf(format,value1,value2,...)
is based on the printf action. The value of sprintf() 
is the string displayed by the action

printf(format,value1,value2,...)

For example,
str = sprintf("%d %d!!!\n",2,3)

assigns the string 2 3!!!\n to the string variable str.

tolower(string) returns the value of string, but with all the letters put into 
lowercase. (This function is an extension to standard awk.)

toupper(string) returns the value of string, but with all the letters put into 
uppercase. (This function is an extension to standard awk.)

ord(string) converts the first character of string into a number. This 
number gives the decimal value of the character in the 
ASCII character set. (This function is an extension to 
standard awk.)

7Arrays

In most programming languages, an array is a list of values, something like 
a table of information. Arrays in awk are more general, but a good start is to 
discuss the traditional concept of an array.

Arrays with Integer Subscripts
The simplest sort of array is a list of values (either numbers or strings). The 
values in the list are called the elements of the array. 

Elements in an array are most commonly referred to by number. For 
example, the first element in the array could be number 1, the second could 
be number 2, and so on. These numbers are called subscripts of the array 
elements.

An awk array has a name, similar to a variable name. To refer to an element 
of an array, you give the name of the array, followed by square brackets 
containing the element's subscript. For example,

arr[3]

refers to element 3 in an array named arr. A statement like
for (i=1; i<=NF; i++)
    arr[i] = $i

creates an array arr with elements that contain all the fields of the current 
record. A program like

{ lines[NR] = $0 }

stores the entire contents of the input file in an array called lines . 
(Remember that the variable NR goes up by 1 for each line that is read in, so 
the elements in the lines array are the lines of the input file, in order.) The 
program

     { lines[NR] = $0 }
END  { for (i=NR; i>0; i--) print lines[i] } 

reads in the contents of a data file and stores the input in lines. When all the 
lines have been read in, the END action displays the lines in reverse order. 
The program therefore reads in lines of text, then displays those lines in 
reverse order. 

Note You would not want to use this program with a large file—awk stores 
arrays in your computer's memory, so there is a limit on how big an array can be.

As another example of the simple use of arrays, suppose you have a file that 
contains 12 columns of numbers and you want to add up the numbers in each 
column. You could do this with the program

     { for (i=1; i<=12; i++) sum[i] = sum[i] + $i }
END  { for (i=1; i<=12; i++) print sum[i] } 

Each element in the array called sum is used to hold a running total of the 
sum of numbers in a particular column.

Notice that our examples have made extensive use of the for statement. 
This is true of many programs that use arrays.

Also notice that you do not need any special statement to create or declare 
an array. If a statement in a program contains a name followed by a value in 
square brackets, awk assumes that the name refers to an array. You cannot 
use a name as both a variable and an array in the same awk program.

Generalized Arrays
Most programming languages let you create arrays that use numbers as 
subscripts; but awk also lets you create arrays that have string values as 
subscripts. As an example, let's go back to the hobbies file and write a 
program that calculates how much each person spends on all his or her 
hobbies.

{ money[$1] += $4 }

uses an array named money. The subscripts of this array are the names of the 
people in the hobbies file. There is

money["Jim"]
money["Linda"]
money["John"]

and so on.



Generalized Arrays

Using MKS AWK 59

Please note that
money[$1] += $4

is equivalent to
money[$1] = money[$1] + $4

except that the expression inside the square brackets is only evaluated once. 
To understand why this is important, consider the following statement: 

This notation is explained in 
“Compound Assignments” on 
page 72.

money[i++] +=2

compared to
money[i++] = money[i++] + 2

In the first case, i only gets incremented once. In the second case, it gets 
incremented twice.

If you apply the program to the input record

Jim     reading          15     100.00
the action becomes

money["Jim"] += 100.00

As with all numeric variables, money["Jim"] starts out with the value 0. At 
the end of the program, the array element contains the amount of money that 
Jim spends on all his hobbies. 

To display the contents of the money array, you can use a new form of the for 
statement:

for (s in money) print s, money[s]

This form of the for statement executes the print action once for every value 
that is used as a subscript for the money array. In each loop, the variable s 
has one of the subscript values; therefore the first time through the loop, s 
might have the value "Jim", the next time "Linda", and so on. The order is 
undefined; therefore the complete program 

     { money[$1] += $4 }
END  { for (s in money) print s, money[s] } 

displays the amount that each person spends on his or her hobbies. You 
should run this program to see how it works. After you have done this, 
replace the print action with a printf to produce more understandable 
output.

Generalized arrays have a wide variety of applications. For example, the 
following program produces a list of all the words used in an input text file. 

     { for (i=1; i<=NF; i++)
           wordlist[$i] = 1 }
END  { for (x in wordlist)
           print x }

Assigning 1 to each element of wordlist is just a dummy action; the 
important thing is that the program creates an element of wordlist with a 
subscript value that is one of the words in the input text file. The for loop in 
the END action then displays all the words that were used as subscript values, 
and this is the set of all words used in the file.

As an exercise, modify the preceding program so that it counts how often 
each word is used in the input file. At the end, the program should display 
each word in the file and how often it was used.

String Subscripts vs. Numeric Subscripts
Previously, we have focussed on numeric subscripts because programmers 
are more familiar with them than string subscripts; however, the truth is that 
awk converts all array subscripts to strings. The result is similar to 
specifying the function

sprintf (CONVFMT, number ...)

where number is the subscript being converted and the CONVFMT variable 
gives the format to use. By default, CONVFMT is %.6g. As a result, when you 
use arr[1], awk converts the subscript to a string; thus you are really using 
arr["1"]:. If you specify arr[1.0], awk converts the subscript to the 
simplest equivalent number (namely 1), and then to the corresponding string 
and you get arr["1"] again. Thus the following are equivalent:

arr[1]   arr[1.0]   arr["1"]

Note arr["01"] may look like it should be equivalent, but it isn't; it's a 
different string.

Deleting Array Elements
Since array elements are all stored in the computer's memory, you can 
decrease memory requirements by deleting elements when you are finished 
using them. To do this, use the statement

delete arrayname[subscript]

as in
delete money["Jim"]

As an extension of standard awk, the statement
delete money

deletes the entire array. It is equivalent to
for (ind in money)
    delete money[ind]

Multi-Dimensional Arrays
awk lets you define arrays with more than one index. Indices are separated 
by commas and enclosed in square brackets, as in

a[1,2] = 3
b["cat", "dog", "bird"] = "horse"

As an example, here is the definition of an array that records different animal 
names.

name["chicken", "male"] = "rooster"
name["chicken", "female"] = "hen"
name["chicken", "young"] = "chick"
name["cattle", "male"] = "bull"
name["cattle", "female"] = "cow"
name["cattle", "young"] = "calf"

As you can see, it is simple to create and manipulate a database that is just an 
awk multi-dimensional array.

8User-Defined 
Functions

Previous chapters discussed numeric functions like sin() and sqrt(), 
and string functions like gsub() and length(). This chapter shows how 
awk lets you create your own functions to perform similar kinds of 
operations.

Function Definition
In an awk program, a function definition looks like this.

function name(argument-list) {
    statements
}

The argument-list is a list of one or more names (separated by commas) that 
represent argument values passed to the function. When an argument name 
is used in the statements of a function, it is replaced by a copy of the 
corresponding argument value.

For example, here is a simple function that takes a single numeric argument 
N and returns a random integer between 1 and N (inclusive). 

function random(N) {
    return (int(N * rand() + 1))
}

This uses two built-in functions discussed in the “Simple Arithmetic” 
chapter: rand() (which returns a random floating point number between 0 
and 1) and int() (which returns the integer part of a floating point number). 
Since rand() returns a floating point number between 0 and 1 (not 
including 1),

N * rand() + 1

is a random floating point number between 1 and N+1 (not including N+1 
itself). Applying the int() function to this floating point number obtains an 
integer between 1 and N. The return statement returns this value as the 
result of the function random().

The definition of random() may appear anywhere in an awk program that a 
normal rule can. A call to a user-defined function can appear in the pattern or 
action part of any rule.

Suppose you have a file that contains people's names in its first field, and 
each of these people is going to roll two six-sided dice. You could simulate 
this situation with the following program. 

function random(N) {
    return (int(N * rand() + 1))
}
{
    score = random(6) + random(6)
    printf "%s rolls %d\n", $1, score
}

This program consists of a definition for the random() function and a rule 
to be applied to every record in the file. The score variable contains the 
sum of two simulated six-sided die rolls. This value is displayed along with 
the name of the person who supposedly rolled the dice.

You can test this program on the hobbies file if you want. Remember, 
however, that the file contains several lines for most people, so the output 
shows more than one roll per person. 

As another example of the random() function, here is the program that was 
used to generate the random baseball scores in the baseball file. The input 
data file had a single line giving the names of baseball teams (separated by 
tabs).

BEGIN  { FS = "\t" }  # Tab is field separator
function random(N) {
    # Produce random number between 1 and N
    return ( int(N * rand() + 1) )
}
{
    # Read in names of baseball teams
    for (i = 1; i <= NF; i++)
        team[i] = $i 
    # Generate 100 random scores
    for (i = 1; i <= 100; i++) { 
        # Choose teams
        hometeam = team[random(NF)]
        visteam = team[random(NF)] 
        # Make sure teams are different

        while (hometeam == visteam)
            visteam = team[random(NF)] 
        # Generate scores
        homescore = random(13)
        visscore = random(13) 
        # Make sure scores are different
        while (homescore == visscore)
            visscore = random(13) 
        # Print out score
        printf "%s\t%d\t",hometeam,homescore
        printf "%s\t%d\n",visteam,visscore
    }
}

The comments in the program should make it easy to understand what is 
happening in each section. Teams were chosen at random from the list of 
teams in the input file, and we made sure we got two different teams. Scores 
were chosen at random between 1 and 13, since this range is typical for 
baseball games. The final results were displayed with two printf 
statements; a single printf statement could have been used, but it would 
have been too wide to fit on this printed page.

As another example of the random() function, here is the program that was 
used to generate the random lists of numbers in the numbers file. 

function random(N) {
    # Produce random integer between 1 and N
    return ( int(N * rand() + 1) )
}
BEGIN {
    for (i = 1; i <= 30; i++) {
        for (j = random(10); j > 0; j--)
            printf "%d ",random(100)
        printf "\n"
    }
    exit
}

Notice that this program only has a BEGIN rule. This rule displays 30 lines, 
each of which contains a random number of integers in the range 1 to 100. 
Notice that this program uses random() to choose the integers, and to 
decide how many of these integers there would be on each line.

Recursion
A function can call itself. This process is called recursion. The usual 
example of a recursive function is the factorial() function.

factorial(N)

is the number which is the product of all integers less than or equal to N. For 
example,

factorial(4)

is 4*3*2*1 or 24. For convenience, the factorial of any N less than 1 is 
considered to be 1.

The following function definition defines the factorial function in a recursive 
way.

function factorial(N) {
    if (N <= 1)
        return 1
    else
        return N * factorial(N-1)
}

If N is less than or equal to 1, the factorial is just 1; otherwise, the factorial of 
N is N times the factorial of N-1. Thus the factorial of 4 (4*3*2*1) is 4 times 
the factorial of 3 (3*2*1). The factorial() function calls itself 
recursively to figure out the appropriate result.

The factorial() function shows that a function may have more than one 
return statement. When a return statement is executed, the function 
immediately stops executing and returns the given value as the function 
result.

Call By Value
When a user-defined function is called, the function receives copies of the 
argument values specified in the function call. For example, suppose a 
program is using a variable named X and calls a user-defined function F() 
with

F(X)

The function F() receives a copy of the current value of X. Because F() 
only has a copy, the function cannot affect the current value of X. For 
example, consider this program.

function exchange(A,B) {
    temp = A
    A = B
    B = temp
}
{
    exchange($1,$2)
    print $0
}

If you look at exchange(), it looks like the function swaps the values of 
arguments A and B. The value of A is temporarily stored in temp; the value of 
B is assigned to A and the saved value of A is assigned to B. Now, when the 
main rule of the program issues the function call

exchange($1,$2)

does awk swap the values of the first two fields of the current record? No. 
The function is only working with copies of the two fields; the function does 
not change the fields themselves.

Notice that the definition of exchange() does not have a return 
statement. It is not necessary for functions to return values. If a function does 
not have a return statement, the function returns when the last statement 
has been executed.

If a function does not use return to return a result, other parts of the 
program should not use that function as if it does return a result. If they try to 
do this, they will get a result value that is meaningless (that is, undefined).

Passing Arrays to Functions
When an array is passed as an argument to a function, it is passed by 
reference. This means that the function works with the actual array, not with 
a copy. Anything that the function does to the array has an effect on the 
original array.

split() is a built-in function that takes an array as an argument. Its form is
split(string,array)

split() breaks up the string into fields, and assigns each of the fields to an 
element of the array. The first field is assigned to array[1], the next to 
array[2], and so on. Fields are assumed to be separated with the field 
separator string FS. If you want to use a different field separator string, you 
can use 

split(string,array,fsstring) 

where fsstring is the field separator string you want to use instead of FS. The 
result of split() is the number of fields that string contained.

Note split() actually changes the elements of array. When an array is 
passed to a function, the function may change the array elements.

9Miscellaneous Topics

In this chapter, we cover a variety of topics which have not yet been 
discussed.

The Getline Function
The getline function reads input from the current data file or from a 
different file. The function has several different forms, discussed in the 
subsections that follow.

Reading from 
the Current 
Input

The simplest form of getline is just
getline

This reads a new record from the current data file. This automatically 
changes the value of $0 and all the other field values. It also changes 
variables like NF, NR, and FNR. In other words, using getline in this way is 
exactly like what happens when awk reads in a new record in the normal 
way. As a simple example,

/xyz/ { print ; getline ; print }

is a rule that prints every record that contains the string xyz. Next, the 
getline reads the next record and the final print prints the record. 
Therefore, the rule prints every record that contains xyz and also the record 
that follows (regardless of what the next record contains).

If you have a program of the form
/xyz/ { print ; getline ; print }
/abc/ { ... some action ... }

the call to getline discards the xyz record and gets a new record. 
Therefore the abc rule will be applied to the new record (if appropriate); it 
will not be applied to the xyz record, because that record is discarded when 
the new record is read.

If a call to getline appears in the BEGIN action, it starts reading the file(s) 
appearing on the command line (if any).

Reading a Line 
into a String 
Variable

The getline function may also be called in the form
getline variable

This reads in a new line from the current data file, but assigns the contents of 
the line to the given string variable. The variables NR and FNR are changed, 
to reflect that another record has been read from the input data file, but the 
contents of $0 and NF are not changed. Therefore a rule like

{
    getline X
    if (X == $0)
        print "Duplicate line"
}

reads a line into the variable X and compares this new line to the old line that 
is still stored in $0.

Reading from a 
New File

Another form of getline reads a line from a new file (not the current data 
file).

getline var <"filename"

reads a line from the given file and stores the contents of the line in the string 
variable var. For example, here is a simple program that compares the 
current data file to another file named testfile and prints out a message if 
the two are not identical.

{
    getline X <"testfile"
    if ($0 != X)
        print "Not identical!"
}

This rule is executed for every line in the data file. Every time the action is 
executed, the getline function reads a new line from testfile and the 
new line is compared to the current line from the data file. For every line 
read from the current data file, another line is read from testfile and the 
two lines are compared. If the two files differ at any point, the message Not 
identical! is printed.

A program may also call getline with the form
getline <"filename"

In this case, a line is read from the given file and assigned to $0. The value 
of NF is changed to reflect the new record in $0, but the variables NR and 
FNR are not changed, because the record wasn't read from the current data 
file.

Reading from 
Other 
Commands

The getline function can also be used to read data produced by another 
command or program.

"command" | getline var

executes the given command and gathers the command's output. The first 
line of output is piped (assigned) into the string variable var. For example,

"date" | getline now

executes the command
date

and assigns the output of the command to the string variable now. For 
instance, the statements

"date" | getline now
if (now ~ /.*Apr.*/)
    print "April Shower Time!"

read the current date into the variable now and checks to see if the date string 
contains Apr.

You can also pipe command output into $0. This is done with a statement of 
the form

"command" | getline

This changes the value of $0 and NF but does not change NR or FNR.

Output to Files and Pipes
Details are given in the Output 
section of the awk reference page 
in the PTC MKS Toolkit Utilities 
Reference.

The output of print and printf may be redirected to a file or a pipe. 

Only a limited number of files and pipes may be opened at one time. The 
close function may be used to close files during execution. In this way, any 
number of files and pipes may be used during the execution of an awk 
program. You can close both input files (used by getline) and output files 
(used by print and printf). 

Executing Programs and System Commands
An earlier section showed how you could execute programs and system 
commands from awk programs using the getline function. You can also 
execute commands with the system function.

system("command line")

executes the given command line. For example,
system("cd XYZ")

executes a cd command to change the current directory.

Compound Assignments
As you have seen, statements like

sum = sum + value

are common in awk programs. For this reason, awk lets you use a short hand 
notation.

sum += value

is exactly the same as
sum = sum + value

but is simpler to write.

The += operation is an example of a compound assignment. Below we list all 
the compound assignment operations of awk and their equivalents.,

Compound 
Operation

Equivalent

A += B A = A + B

A -= B A = A - B

A *= B A = A * B

A /= B A = A / B

A %= B A = A % B

A ^= B A = A ^ B

As an example
/John/ { sum += $3 }

is a program we could use on our hobbies file to calculate how many hours 
a week John spends on his hobbies.

The SORTGEN Example
In the Basics of the PTC MKS Toolkit document, the sort command was 
used as part of an example. There, it was mentioned how hard it can be to 
remember all of the options to sort, and described an awk program called 
sortgen that would generate the correct options for a specification.

Briefly, sortgen takes a description of the layout of the fields in a record, 
and emits a command line for sort that carries out the desired sort. Note 
that sortgen uses 1-origin (the first field to be sorted on is field 1), and 
writes the sort command line to use sort's 0-origin field labelling. Here is 
the definition of sortgen.
# sortgen - generate sort command
#   input: sequence of lines describing sort options
#   output: command line for sort 
BEGIN { key = 0 } 

/no |not |n't / {
print "error: cannot do negatives:", $0; ok = 1

} 
# rules for global variables
{ ok = 0 }
/uniq|discard.*(iden|dupl)/ { uniq = " -u"; ok = 1 }
/separ.*tab|tab.*separ/ { sep = "t'\t'"; ok = 1 }
/separ/  { for (i = 1; i <= NF; i++)

if (length($i) == 1)
sep = "t'" $i "'"

  ok = 1
}
/key/ { key++; dokey(); ok = 1 }# new key; must come in order 
# rules for each key 
/dict/  { dict[key] = "d"; ok = 1 }
/ignore.*(space|blank)/ { blank[key] = "b"; ok = 1 }
/fold|case/  { fold[key] = "f"; ok = 1 }
/num/  { num[key] = "n"; ok = 1 }
/rev|descend|decreas|down|oppos/ { rev[key] = "r"; ok = 1 }
/month/  { month[key] = "M"; ok = 1 }
/forward|ascend|increas|up|alpha/ { next }# this is default
!ok  { print "error: cannot understand:", $0 } 
END  { # print flags for each key

cmd = "sort" uniq

flag = dict[0] blank[0] fold[0] rev[0] num[0] month[0] 
sep

if (flag) cmd = cmd " -" flag
for (i = 1; i <= key; i++)

if (pos[i] != "") {
flag = pos[i] dict[i] blank[i] fold[i]
flag = flag rev[i] num[i] month[i]
if (flag) cmd = cmd " -k" flag
if (pos2[i]) cmd = cmd "," pos2[i]

}
print cmd

}
function dokey(i) {# determine position of key

for (i = 1; i <= NF; i++)
if ($i ~ /^[0-9]+$/) {

pos[key] = $i
break

}
for (i++; i <= NF; i++)

if ($i ~ /^[0-9]+$/) {
pos2[key] = $i + 1
break

}
if (pos[key] == "")

printf("error: invalid key specification: %s\n", $0)
if (pos2[key] == "")

pos2[key] = pos[key]
}

Running awk Programs Without awk
For more information about 
awkc, see the awkc reference 
page in the PTC MKS Toolkit 
Utilities Reference.

Sometimes you want to run an awk program as a separate program, without 
running awk. Perhaps you want to give a useful program to a friend who 
doesn't own awk. PTC MKS Toolkit for Developers, PTC MKS Toolkit for 
Interoperability, PTC MKS Toolkit for Professional Developers, and PTC 
MKS Toolkit for Enterprise Developers includes an awk compiler, awkc, 
which converts awk source programs into standalone programs.

awkc works like awk, and accepts many of the same options. To convert the 
awk source file sortgen into the executable sortgen.exe, type:

awkc -o sortgen.exe -f sortgen

The -o option specifies the name of the output executable file; you must 
specify a name. Remember that executable files have the .exe extension.

If you do not specify a program source file, awkc waits for the program from 
the command line, just like awk. For example, if you want to create a simple 
program called lineno.exe to display the contents of a file with line 
numbers, enter:

awkc -o lineno.exe '{print NR ":" $0}'

